Приветствую)

Нужен отчет, который выводить данные продаж по Подразделению и относящимся к этому подразделению Менеджеров (развития и операционный). Загвоздка вот в чём - надо как то привязать этих манагеров к подразделению, без изменения конфигурации, сделал через "Значение свойств объектов", но при группировке в отчете они не встают вместе, а разбиваются.

Вопрос есть ли способ это дело поравить или привязать как то по другому?

Текст запроса:

```1c
    ВЫБРАТЬ
    ВТ.Период,
    ВТ.Подразделение,
    ВТ.Номенклатура,
    СУММА(ВТ.КоличествоПлан) КАК КоличествоПлан,
    СУММА(ВТ.СуммаПлан) КАК СуммаПлан,
    СУММА(ВТ.КоличествоФакт) КАК КоличествоФакт,
    СУММА(ВТ.СуммаФакт) КАК СуммаФакт,
    ВЫБОР
    КОГДА ЗначенияСвойствОбъектов.Объект = ВТ.Подразделение
    И ЗначенияСвойствОбъектов.Свойство = &Менеджер
    ТОГДА ЗначенияСвойствОбъектов.Значение
    КОНЕЦ КАК Менеджер, //<<<<<<<<<<<<<<<<<<<<<< один товарищ
    ВЫБОР
    КОГДА ЗначенияСвойствОбъектов.Объект = ВТ.Подразделение
    И ЗначенияСвойствОбъектов.Свойство = &Операционист
    ТОГДА ЗначенияСвойствОбъектов.Значение
    КОНЕЦ КАК Операционист //<<<<<<<<<<<<<<<<<<<<<< второй товарищ
    ИЗ
    ВТ КАК ВТ
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЗначенияСвойствОбъектов КАК ЗначенияСвойствОбъектов
    ПО ВТ.Подразделение = ЗначенияСвойствОбъектов.Объект
```

на рисунке наглядно показал результат
---
Две записи в регистре сведений - две записи и в отчете. Группировка в Вашем случае ничем не поможет, т.к. в одной строке менеджер пустой, операционист заполнен, а в другой строке - ровно наоборот. Т.е. две уникальных пары значений.

Проще всего использовать две связи с регистром дополнительных значений: одна связь для получения менеджера, другая - для операциониста:

```1c
    ВЫБРАТЬ
    ВТ.Период,
    ВТ.Подразделение,
    ВТ.Номенклатура,
    ВТ.КоличествоПлан КАК КоличествоПлан,
    ВТ.СуммаПлан КАК СуммаПлан,
    ВТ.КоличествоФакт КАК КоличествоФакт,
    ВТ.СуммаФакт КАК СуммаФакт,
    СвойствоМенеджер.Значение КАК Менеджер,
    СвойствоОперационист.Значение КАК Операционист
    ИЗ
    ВТ КАК ВТ
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЗначенияСвойствОбъектов КАК СвойствоМенеджер
    ПО ВТ.Подразделение = СвойствоМенеджер.Объект
    И СвойствоМенеджер.Свойство = &Менеджер
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЗначенияСвойствОбъектов КАК СвойствоОперационист
    ПО ВТ.Подразделение = СвойствоОперационист.Объект
    И СвойствоОперационист.Свойство = &Операционист
```
---
Две записи в регистре сведений - две записи и в отчете. Группировка в Вашем случае ничем не поможет, т.к. в одной строке менеджер пустой, операционист заполнен, а в другой строке - ровно наоборот. Т.е. две уникальных пары значений.

Проще всего использовать две связи с регистром дополнительных значений: одна связь для получения менеджера, другая - для операциониста:

```1c
    ВЫБРАТЬ
    ВТ.Период,
    ВТ.Подразделение,
    ВТ.Номенклатура,
    ВТ.КоличествоПлан КАК КоличествоПлан,
    ВТ.СуммаПлан КАК СуммаПлан,
    ВТ.КоличествоФакт КАК КоличествоФакт,
    ВТ.СуммаФакт КАК СуммаФакт,
    СвойствоМенеджер.Значение КАК Менеджер,
    СвойствоОперационист.Значение КАК Операционист
    ИЗ
    ВТ КАК ВТ
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЗначенияСвойствОбъектов КАК СвойствоМенеджер
    ПО ВТ.Подразделение = СвойствоМенеджер.Объект
    И СвойствоМенеджер.Свойство = &Менеджер
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЗначенияСвойствОбъектов КАК СвойствоОперационист
    ПО ВТ.Подразделение = СвойствоОперационист.Объект
    И СвойствоОперационист.Свойство = &Операционист
```
---
примного благодарен))
---
На 2 сделать 2 соединения с РС - один для менеджера, другой для операциониста